

Salesforce Marketing Cloud

AMPscript IF Statements

Exercise 1



Overview

This exercise is designed to reinforce your learnings and test your understanding of AMPscript IF Statements.

Process the tasks in order to complete this exercise.

Each Task in the following Brief will get progressively harder, so if you get stuck or are unsure how to proceed, you can join the conversation in the comments section here: <https://youtu.be/ybucW4cCxME>

The last pages of this document contain some solutions. Make sure you try solve them yourself before jumping to the end.

I hope you enjoy this exercise and it helps you to improve your skills with AMPscript IF Statements!

Required Files

This exercise requires the use of a Data Extension with some sample data.

You can copy/download all of the items you need from this exercise from this GitHub resource:

<https://github.com/camrobert/SalesforceMarketingCloud/tree/main/Challenge/AMP-E01>



Brief


You work in a small e-commerce company that uses Marketing Cloud. During your day you get the following requests from the Marketing Manager in relation to an upcoming email send for an upcoming product launch.


Email Setup


- 1 In Content Builder, Create an email using the "Blank Page" template. Name your email "AMP-E01 <yourname>"
- 2 Drag a HTML Content Block onto the email and paste the following text into it:


Hi %%Firstname%%,
We are excited to announce our new product lineup will be launching next month.

The email should now look like this:

Subject	AMP-E01
Preheader	 Add Preheader

 Default

 Plain Text

 Code View

Hi %%Firstname%%,
We are excited to announce our new product lineup will be launching next month.



Task 1

The Marketing Manager has asked for ideas on how to improve results for the upcoming campaign. Having recently read the Litmus State of Email report, you suggest to add some more personalisation to the email by using the customer's Preferred Name (if they have one).

The Marketing Manager loves it, and asks you to implement it right away!

Objective

Add AMPScript to the email so that it uses the Customer's [PreferredName] if they have one, otherwise it should default to their [Firstname].

Expected Results

Subscriber Record

AMPscriptIFExample_01
AMPscriptIFExample_02
AMPscriptIFExample_03
AMPscriptIFExample_04

Text First Line:

"Hi Ken,"
"Hi Fifi,"
"Hi Beryl,"
"Hi Gaz,"



Task 2

The Content Team has revised the introduction paragraph to include a reference to the customer's loyalty tier:

|| We are excited to announce our new product lineup will be launching ||
next month. As a valued %%Tier%% Tier Member, we want to ensure
you continue to get access to the best products and deals.

You advise the Content Team that there is no "Tier" value in the data, and they provide you with the following table to "just make it work"...

Points Range	Tier Value
0 - 99	Bronze
100 - 249	Silver
250+	Gold

Objective

Write some AMPscript to produce the correct Tier value based on the Customer's [PointsYTD] value.

Expected Results

Subscriber Record

AMPscriptIFExample_01
AMPscriptIFExample_04
AMPscriptIFExample_10

Text First Line:

... valued Bronze Tier...
... valued Silver Tier...
... valued Gold Tier...



Task 3

The Marketing Manager wants to include some copy to address when the customer last transacted with the company.

The following 2 sentences have been provided by the Manager:

Segment: Recent Purchase

|| We are grateful for your continued interest in our products. ||

Segment: No Recent Purchase

|| We hope to see you enjoying our products again soon. ||

You are told that only 1 sentence should show for each customer, and the text should appear directly after the existing copy.

A customer who has made a purchase in the last 14 days is "Recent".

Objective

Create a new HTML Content Block and write some AMPscript to use the value in [LastPurchaseDays] to determine which message should show. Check the Data for any issues.

Expected Results

Subscriber Record

AMPscriptIFExample_01
AMPscriptIFExample_02
AMPscriptIFExample_03

Show which Text:

Not Recent
Recent
Not Recent



Task 4

The E-Commerce Team has developed a promotional offer that will give a 5% discount to high-purchase-count customers (5 or more purchases) or have made a purchase in the last 7 days.

The Marketing Manager has approved the inclusion of some text to advise the customer that are eligible for a 5% discount:

|| As a loyal customer, we would like to offer you a further 5% discount on your next purchase on our website. Use promo code "Thanks5" on checkout. ||

The Manager wants the text to appear in a separate Content Block, directly below the existing text.

Objective

Create a new HTML Content Block and write some AMPscript to show/hide the text provided based on their [PurchasesYTD] or [LastPurchaseDays] value.

Expected Results

Subscriber Record

AMPscriptIFExample_01
AMPscriptIFExample_02
AMPscriptIFExample_10

Show Text:

Hide
Show
Show



Task 5

The BI Team has identified shipping fees to be a major blocker for customers. Some customers are already eligible for Free Shipping, however they have created some additional criteria to offer a Free Shipping code in the email:

Any customer who joined in the Last 90 days, or have over 200 Points, or have spent over \$1000 Year to Date.



GET FREE SHIPPING ON YOUR NEXT ORDER!
Use code "0Shipping" on checkout!



Objective

Create a new HTML Content Block and use AMPscript to show the provided text for customers who already have [FreeShipping] = TRUE, or that meet the provided criteria using [JoinAgeDays], [PointsYTD] and [SpendYTD].

Expected Results

AMPscriptIFExample_13 is the only customer who will NOT have Free Shipping.



Task 6

The Marketing Team would like to cross-promote a preference in the email for any customers who are not already signed up to it. They have provided 3 sets of copy based on their marketing preferences:

#1: FALSE to [Marketing_FlashSales], FALSE to [Marketing_Campaigns]

|| You can stay up to date with our best sales by subscribing to our Flash Sales announcements, and get seasonal deals with our Campaigns Newsletter. ||

#2: FALSE to [Marketing_FlashSales], TRUE to [Marketing_Campaigns]

|| You can stay up to date with our best sales by subscribing to our Flash Sales announcements. ||

#3: TRUE to [Marketing_FlashSales], FALSE to [Marketing_Campaigns]

|| You can get access to great seasonal deals with our Campaigns Newsletter. Update your preferences to Subscribe! ||

Objective

Create a new HTML Content Block and use AMPscript to show the correct message based on each customer's marketing flags. Don't show any message if they are subscribed to both preferences.

Expected Results

Subscriber Record

AMPscriptIFExample_02
AMPscriptIFExample_05
AMPscriptIFExample_15
AMPscriptIFExample_01

Show which Text:

No Text
#3 Text
#2 Text
#1 Text



Task 7

The Email is almost ready to go when the PR team advises you they want to hold a VIP Launch event for select customers.

They give you the following Text and Criteria:

|| To celebrate the launch of our new product, we would like to invite you to a Launch Party. Please contact your Sales Representative to secure your spot. ||

All Current Gold Tier customers, plus any Customers who will likely reach Gold Tier based on their YTD Points earnings with relation to their Review date.

Objective

Create a new HTML Content Block and use AMPscript to show the provided text for customers who meet the criteria. The BI Team has advised that you can calculate if a customer will reach Gold Tier by projecting their points balance over the remaining time until their next review using the following equation:

$$\frac{[PointsYTD]}{\left(\frac{365 - [NextReviewDays]}{365} \right)}$$

Expected Results

Subscriber Record

AMPscriptIFExample_10
AMPscriptIFExample_08
AMPscriptIFExample_06

Calculation Outcome:

Invite (Current Gold Tier)
Invite (Projected Gold Tier)
No Invite



Answers



Answers

Note: Many of these Tasks can be solved multiple ways using AMPscript. The Answers below are not exclusively correct, however they will work if copied into your email for this exercise.

Answers are shown in HTML for formatting purposes.

1

```
%%[  
IF EMPTY(PreferredName) THEN  
SET @name = Firstname  
ELSE  
SET @name = PreferredName  
ENDIF  
]%%  
Hi %%=v(@name)=%%,<br>  
We are excited to announce our new product lineup will be launching  
next month.
```

BONUS Solution

Common Solution using an Inline IF Statement

```
Hi %%=IIF(EMPTY(PreferredName),Firstname,PreferredName)=%%,<br>  
We are excited to announce our new product lineup will be launching next  
month.
```



Answers

2

```
%%[  
IF NOT EMPTY(PreferredName) THEN  
SET @name = PreferredName  
ELSE  
SET @name = Firstname  
ENDIF
```

```
IF PointsYTD >= 250 THEN  
SET @tier = "Gold"  
ELSEIF PointsYTD >= 100 THEN  
SET @tier = "Silver"  
ELSE  
SET @tier = "Bronze"  
ENDIF  
]%%
```

Hi %%=v(@name)=%%,

We are excited to announce our new product lineup will be launching next month. As a valued %%=v(@tier)=%% Tier Member, we want to ensure you continue to get access to the best products and deals.

3

```
%%[IF LastPurchaseDays > 0 AND LastPurchaseDays <= 14 THEN]%%  
We are grateful for your continued interest in our products  
%%[ELSE]%%  
We hope to see you enjoying our products again soon.  
%%[ENDIF]%%
```



Answers

4

```
%%[IF PurchasesYTD >= 5 OR LastPurchaseDays <= 7 THEN]%%
```

As a loyal customer, we would like to offer you a further 5% discount on your next purchase on our website. Use promo code "Thanks5" on checkout.

```
%%[ENDIF]%%
```

5

```
%%[IF FreeShipping OR JoinAgeDays < 90 OR PointsYTD >= 200 OR  
SpendYTD >= 1000 THEN]%%
```

GET FREE SHIPPING ON YOUR NEXT ORDER!

Use code "0Shipping" on checkout!

```
%%[ENDIF]%%
```

6

```
%%[IF NOT Marketing_FlashSales AND NOT Marketing_Campaigns  
THEN]%%
```

You can stay up to date with our best sales by subscribing to our Flash Sales announcements, and get seasonal deals with our Campaigns Newsletter.

```
%%[ELSEIF NOT Marketing_FlashSales AND Marketing_Campaigns  
THEN]%%
```

You can stay up to date with our best sales by subscribing to our Flash Sales

```
%%[ELSEIF Marketing_FlashSales AND NOT Marketing_Campaigns  
THEN]%%
```

You can get access to great seasonal deals with our Campaigns Newsletter. Update your preferences to Subscribe!

```
%%[ENDIF]%%
```



Answers

7

```
%%[IF PointsYTD >= 250 OR  
Divide(PointsYTD,Divide(Subtract(365,NextReviewDays),365)) >= 250  
THEN]%%
```

To celebrate the launch of our new product, we would like to invite you to a Launch Party. Please contact your Sales Representative to secure your spot.

```
%%[ENDIF]%%
```

